



Pendle v2 (Part 2) Audit Report

Aug 2, 2022





Table of Contents

Summary	2
Overview	3
Issues	4
WP-M1: Wrong implementation of <code>PendleVotingControllerUpg#vote()</code> will revert when there are pools that get more weights than before	4
WP-L2: Duplicate event emissions	9
WP-I3: Expired markets should be excluded from PENDLE rewards automatically	11
WP-I4: It's possible that the pool address can be the same on different networks, and we should avoid that	13
WP-L5: <code>PendleVotingControllerUpg.sol#_broadcastResults()</code> Precision loss due to <code>div</code> before <code>mul</code>	16
WP-I6: The external rewards should not be distributed according to vePENDLE shares based <code>activeBalance</code>	17
WP-G7: <code>vePENDLE.sol</code> Combine two external calls into one can save gas	19
WP-G8: Avoid unnecessary storage read can save gas	20
WP-I9: End the <code>Vote</code> a few hours earlier before the next week starts can help avoid leftover	21
WP-I10: <code>VotingEscrowPendleMainchain#_broadcast()</code> should refund the unspent crosschain message fee	23
WP-I11: Adding mainchain <code>gaugeController</code> into <code>sidechainContracts</code> is confusing and unnecessary	26
WP-I12: Users will not naturally update others' <i>activeBalance</i>	29
Appendix	32
Disclaimer	33



Summary

This report has been prepared for Pendle v2 (Part 2) Audit Report smart contract, to discover issues and vulnerabilities in the source code of their Smart Contract as well as any contract dependencies that were not part of an officially recognized library. A comprehensive examination has been performed, utilizing Static Analysis and Manual Review techniques.

The auditing process pays special attention to the following considerations:

- Testing the smart contracts against both common and uncommon attack vectors.
- Assessing the codebase to ensure compliance with current best practices and industry standards.
- Ensuring contract logic meets the specifications and intentions of the client.
- Cross referencing contract structure and implementation against similar smart contracts produced by industry leaders.
- Thorough line-by-line manual review of the entire codebase by industry experts.



Overview

Project Summary

Project Name	Pendle v2
Codebase	https://github.com/pendle-finance/pendle-core-internal-v2
Commit	6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe
Language	Solidity

Audit Summary

Delivery Date	Aug 2, 2022
Audit Methodology	Static Analysis, Manual Review
Total Issues	12



WP-M1: Wrong implementation of `PendleVotingControllerUpg#vote()` will revert when there are pools that get more weights than before

Medium

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/VotingController/PendleVotingControllerUpg.sol#L69-L91>

```
69  function vote(address[] calldata pools, uint64[] calldata weights) external {
70      address user = msg.sender;
71
72      require(weights.length == pools.length, "invalid array length");
73      require(vePendle.balanceOf(user) > 0, "zero vependle balance");
74
75      UserData storage uData = userData[user];
76      LockedPosition memory userPosition = _getUserVePendlePosition(user);
77
78      for (uint256 i = 0; i < pools.length; ++i) {
79          if (_isPoolActive(pools[i])) applyPoolSlopeChanges(pools[i]);
80      }
81
82      for (uint256 i = 0; i < pools.length; ++i) {
83          if (uData.voteForPools[pools[i]].weight <= weights[i])
84              _modifyVoteWeight(user, pools[i], userPosition, weights[i]);
85      }
86
87      for (uint256 i = 0; i < pools.length; ++i) {
88          if (uData.voteForPools[pools[i]].weight > weights[i])
89              _modifyVoteWeight(user, pools[i], userPosition, weights[i]);
90      }
91  }
```

At L82-85, the pools that get more weights than before will be updated first, which we believe is a wrong implementation; it seems the original intention is to update the pools that get fewer



weights first.

As a result, the transaction will revert at L212-216 with the error: "exceeded max weight".

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/VotingController/VotingControllerStorageUpg.sol#L181-L220>

```
181  function _modifyVoteWeight(  
182      address user,  
183      address pool,  
184      LockedPosition memory userPosition,  
185      uint64 weight  
186  ) internal returns (VeBalance memory newVote) {  
187      UserData storage uData = userData[user];  
188      PoolData storage pData = poolData[pool];  
189  
190      VeBalance memory oldVote = uData.voteForPools[pool].vote;  
191  
192      // REMOVE OLD VOTE  
193      if (oldVote.bias != 0) {  
194          if (_isPoolActive(pool) && _isVoteActive(oldVote)) {  
195              pData.totalVote = pData.totalVote.sub(oldVote);  
196              pData.slopeChanges[oldVote.getExpiry()] -= oldVote.slope;  
197          }  
198          uData.totalVotedWeight -= uData.voteForPools[pool].weight;  
199          delete uData.voteForPools[pool];  
200      }  
201  
202      // ADD NEW VOTE  
203      if (weight != 0) {  
204          require(_isPoolActive(pool), "pool not active");  
205  
206          newVote = userPosition.convertToVeBalance(weight);  
207  
208          pData.totalVote = pData.totalVote.add(newVote);  
209          pData.slopeChanges[newVote.getExpiry()] += newVote.slope;  
210  
211          uData.voteForPools[pool] = UserPoolData(weight, newVote);  
212          uData.totalVotedWeight += weight;  
213          require(  
214              uData.totalVotedWeight <= VeBalanceLib.USER_VOTE_MAX_WEIGHT,  
215              "exceeded max weight"
```



```
216     );  
217   }  
218  
219   userPoolHistory[user][pool].push(newVote);  
220 }
```

PoC

Given:

- Alice allocated weights as such: [pool1: 0.5, pool2: 0.5]

When:

- Alice `vote()` with: [pool1: 0.4, pool2: 0.6]

Then:

- `PendleVotingControllerUpg#vote()` the first iteration at L82-L85 will be skipped as `if (0.5 <= 0.4)` failed
- `PendleVotingControllerUpg#vote()` the second iteration at L82-L85 will call `_modifyVoteWeight(user1, pool2, userPosition, 0.6)` as `if (0.5 <= 0.6)` passed:
 - `_modifyVoteWeight()` L198, `uData.totalVotedWeight -= 0.5;` updated `uData.totalVotedWeight` to `0.5`
 - `_modifyVoteWeight()` L212, `uData.totalVotedWeight += 0.6;` updated `uData.totalVotedWeight` to `1.1`
 - `_modifyVoteWeight()` L213-L216, reverted because of `require(1.1 <= 1.0, "exceeded max weight");`

Recommendation

- Moving the `uData.totalVotedWeight <= VeBalanceLib.USER_VOTE_MAX_WEIGHT` check in `VotingControllerStorageUpg#_modifyVoteWeight()` to the end of the `PendleVotingControllerUpg#vote()` ;
- Merging the two for-loop in `PendleVotingControllerUpg#vote()` into one:



```
69 function vote(address[] calldata pools, uint64[] calldata weights) external {
70     address user = msg.sender;
71
72     require(weights.length == pools.length, "invalid array length");
73     require(vePendle.balanceOf(user) > 0, "zero vependle balance");
74
75     UserData storage uData = userData[user];
76     LockedPosition memory userPosition = _getUserVePendlePosition(user);
77
78     for (uint256 i = 0; i < pools.length; ++i) {
79         if (_isPoolActive(pools[i])) applyPoolSlopeChanges(pools[i]);
80     }
81
82     for (uint256 i = 0; i < pools.length; ++i) {
83         _modifyVoteWeight(user, pools[i], userPosition, weights[i]);
84     }
85
86     require(
87         uData.totalVotedWeight <= VeBalanceLib.USER_VOTE_MAX_WEIGHT,
88         "exceeded max weight"
89     );
90 }
```

```
181 function _modifyVoteWeight(
182     address user,
183     address pool,
184     LockedPosition memory userPosition,
185     uint64 weight
186 ) internal returns (VeBalance memory newVote) {
187     UserData storage uData = userData[user];
188     PoolData storage pData = poolData[pool];
189
190     VeBalance memory oldVote = uData.voteForPools[pool].vote;
191
192     // REMOVE OLD VOTE
193     if (oldVote.bias != 0) {
194         if (_isPoolActive(pool) && _isVoteActive(oldVote)) {
195             pData.totalVote = pData.totalVote.sub(oldVote);
196             pData.slopeChanges[oldVote.getExpiry()] -= oldVote.slope;
197         }
198         uData.totalVotedWeight -= uData.voteForPools[pool].weight;
```



```
199         delete uData.voteForPools[pool];
200     }
201
202     // ADD NEW VOTE
203     if (weight != 0) {
204         require(!_isPoolActive(pool), "pool not active");
205
206         newVote = userPosition.convertToVeBalance(weight);
207
208         pData.totalVote = pData.totalVote.add(newVote);
209         pData.slopeChanges[newVote.getExpiry()] += newVote.slope;
210
211         uData.voteForPools[pool] = UserPoolData(weight, newVote);
212         uData.totalVotedWeight += weight;
213     }
214
215     userPoolHistory[user][pool].push(newVote);
216 }
```

Status

✓ Fixed



WP-L2: Duplicate event emissions

Low

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/VotingEscrow/VotingEscrowPendleMainchain.sol#L213-L231>

```
213     function _broadcastPosition(address user, uint256[] calldata chainIds) public
        payable {
214         require(chainIds.length != 0, "empty chainIds");
215
216         (VeBalance memory supply, uint256 wTime) = _applySlopeChange();
217
218         bytes memory userData = (
219             user == address(0) ? EMPTY_BYTES : abi.encode(user,
positionData[user])
220         );
221
222         for (uint256 i = 0; i < chainIds.length; ++i) {
223             require(sidechainContracts.contains(chainIds[i]), "not supported
chain");
224             _broadcast(chainIds[i], wTime, supply, userData);
225             if (user != address(0)) {
226                 emit BroadcastUserPosition(user, chainIds);
227             }
228         }
229
230         emit BroadcastTotalSupply(supply, chainIds);
231     }
```

`BroadcastUserPosition` will be emitted `chainIds.length` times with all the `chainIds` each time.



Recommendation

```
213     function _broadcastPosition(address user, uint256[] calldata chainIds) public
payable {
214         require(chainIds.length != 0, "empty chainIds");
215
216         (VeBalance memory supply, uint256 wTime) = _applySlopeChange();
217
218         bytes memory userData = (
219             user == address(0) ? EMPTY_BYTES : abi.encode(user,
positionData[user])
220         );
221
222         for (uint256 i = 0; i < chainIds.length; ++i) {
223             require(sidechainContracts.contains(chainIds[i]), "not supported
chain");
224             _broadcast(chainIds[i], wTime, supply, userData);
225         }
226
227         if (user != address(0)) {
228             emit BroadcastUserPosition(user, chainIds);
229         }
230         emit BroadcastTotalSupply(supply, chainIds);
231     }
```

Status

✓ Fixed



WP-I3: Expired markets should be excluded from PENDLE rewards automatically

Informational

Issue Description

Expired markets should no longer receive any PENDLE rewards as the market is usually no longer needed by then.

Per the README.md:

markets that are expired will be removed by governance

While this gives us more flexibility, we also believe it is prone to delay/mistake more than an automatic method.

Recommendation

Consider adding a check in `_receiveVotingResults` and only `_addRewardsToMarket` when the market is not expired. The `pendleAmounts` allocated to the expired markets can be sent to the treasury.

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/GaugeController/PendleGaugeControllerBaseUpg.sol#L93-L109>

```
93  function _receiveVotingResults(  
94      uint128 wTime,  
95      address[] memory markets,  
96      uint256[] memory pendleAmounts  
97  ) internal {  
98      require(markets.length == pendleAmounts.length, "invalid markets length");  
99  
100     if (epochRewardReceived[wTime]) return; // only accept the first message for  
        the wTime  
101     epochRewardReceived[wTime] = true;  
102  
103     for (uint256 i = 0; i < markets.length; ++i) {
```



```
104     _addRewardsToMarket(markets[i], pendleAmounts[i].Uint128());
105 }
106
107     emit ReceiveVotingResults(wTime, markets, pendleAmounts);
108 }
```

If we want a more precise time to end the rewards for the soon-to-expire markets, `_addRewardsToMarket()` can be changed to replace `WEEK` with the length of time until it expires.

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/GaugeController/PendleGaugeControllerBaseUpg.sol#L117-L128>

```
117 function _addRewardsToMarket(address market, uint128 pendleAmount) internal {
118     MarketRewardData memory rwd = _getUpdatedMarketReward(market);
119     uint128 leftover = (rwd.incentiveEndsAt - rwd.lastUpdated) *
rwd.pendlePerSec;
120     uint128 newSpeed = (leftover + pendleAmount) / WEEK;
121
122     rewardData[market] = MarketRewardData({
123         pendlePerSec: newSpeed,
124         accumulatedPendle: rwd.accumulatedPendle,
125         lastUpdated: uint128(block.timestamp),
126         incentiveEndsAt: uint128(block.timestamp) + WEEK
127     });
128 }
```

Status

✓ Fixed



WP-I4: It's possible that the pool address can be the same on different networks, and we should avoid that

Informational

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/VotingController/VotingControllerStorageUpg.sol#L136-L157>

```
136 function _addPool(uint64 chainId, address pool) internal {
137     require(chainPools[chainId].add(pool), "IE");
138     require(allActivePools.add(pool), "IE");
139
140     poolData[pool].chainId = chainId;
141     poolData[pool].lastSlopeChangeAppliedAt = WeekMath.getCurrentWeekStart();
142 }
143
144 /**
145  * @dev expected behavior:
146  * - remove from allActivePool, chainPools
147  * - add to allRemovedPools
148  * - clear all params in poolData
149  */
150 function _removePool(address pool) internal {
151     uint64 chainId = poolData[pool].chainId;
152     require(chainPools[chainId].remove(pool), "IE");
153     require(allActivePools.remove(pool), "IE");
154     require(allRemovedPools.add(pool), "IE");
155
156     delete poolData[pool];
157 }
```

The address of the pool is used as the ID for the all the markets cross the networks, it assumes that the address is unique.

However, We find that it's possible for the address of a pool to be the same on another network, this can be troublesome, so we should avoid that.



<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/Market/PendleMarketFactory.sol#L76-L95>

```
76     function createNewMarket(  
77         address PT,  
78         int256 scalarRoot,  
79         int256 initialAnchor  
80     ) external returns (address market) {  
81         require(IPYieldContractFactory(yieldContractFactory).isPT(PT), "Invalid  
PT");  
82         require(markets[PT][scalarRoot][initialAnchor] == address(0), "market  
already created");  
83  
84         // no need salt since market's existence has been checked before hand  
85         market = SSTORE2Deployer.create2(  
86             marketCreationCodePointer,  
87             bytes32(0),  
88             abi.encode(PT, scalarRoot, initialAnchor, vePendle, gaugeController)  
89         );  
90  
91         markets[PT][scalarRoot][initialAnchor] = market;  
92         require(allMarkets.add(market), "IE market can't be added");  
93  
94         emit CreateNewMarket(market, PT, scalarRoot, initialAnchor);  
95     }
```

When the `PendleMarketFactory` is deployed by the same deployer with the same nonce on a different network, AND when `createNewMarket` is called with the same params, then the pool address will be the same on different networks.

This is unlikely to happen in practice, but still possible.

Recommendation

Consider using the `chainId` as the salt in `createNewMarket()` :

```
76     function createNewMarket(  
77         address PT,  
78         int256 scalarRoot,
```



```
79     int256 initialAnchor
80 ) external returns (address market) {
81     require(IPYieldContractFactory(yieldContractFactory).isPT(PT), "Invalid PT");
82     require(markets[PT][scalarRoot][initialAnchor] == address(0), "market already
    created");
83
84     // no need salt since market's existence has been checked before hand
85     market = SSTORE2Deployer.create2(
86         marketCreationCodePointer,
87         bytes32(block.chainId),
88         abi.encode(PT, scalarRoot, initialAnchor, vePendle, gaugeController)
89     );
90
91     markets[PT][scalarRoot][initialAnchor] = market;
92     require(allMarkets.add(market), "IE market can't be added");
93
94     emit CreateNewMarket(market, PT, scalarRoot, initialAnchor);
95 }
```

Status

✓ Fixed



WP-L5: PendleVotingControllerUpg.sol#_broadcastResults()

Precision loss due to `div` before `mul`

Low

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/VotingController/PendleVotingControllerUpg.sol#L245-L249>

```
245  for (uint256 i = 0; i < length; ++i) {
246      uint256 poolVotes = weekData[wTime].poolVotes[pools[i]];
247      uint256 pendlePerSec = (totalPendlePerSec * poolVotes) / totalVotes;
248      totalPendleAmounts[i] = pendlePerSec * WEEK;
249  }
```

Recommendation

Change to:

```
245  for (uint256 i = 0; i < length; ++i) {
246      uint256 poolVotes = weekData[wTime].poolVotes[pools[i]];
247      totalPendleAmounts[i] = (totalPendlePerSec * poolVotes) * WEEK / totalVotes;
248  }
```

Status

✓ Fixed



WP-I6: The external rewards should not be distributed according to vePENDLE shares based `activeBalance`

Informational

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/PendleGauge.sol#L66-L94>

```
66  function _updateUserActiveBalancePrivate(address user) private {
67      assert(user != address(0) && user != address(this));
68
69      uint256 lpBalance = _stakedBalance(user);
70      uint256 veBoostedLpBalance = _calcVeBoostedLpBalance(user, lpBalance);
71
72      uint256 newActiveBalance = Math.min(veBoostedLpBalance, lpBalance);
73
74      totalActiveSupply = totalActiveSupply - activeBalance[user] +
        newActiveBalance;
75      activeBalance[user] = newActiveBalance;
76  }
77
78  function _calcVeBoostedLpBalance(address user, uint256 lpBalance)
79      internal
80      virtual
81      returns (uint256)
82  {
83      uint256 vePendleBalance = vePENDLE.balanceOf(user);
84      uint256 vePendleSupply = vePENDLE.totalSupplyCurrent();
85      // Inspired by Curve's Gauge
86      uint256 veBoostedLpBalance = (lpBalance * TOKENLESS_PRODUCTION) / 100;
87      if (vePendleSupply > 0) {
88          veBoostedLpBalance +=
89              (((_totalStaked() * vePendleBalance) / vePendleSupply) *
90              (100 - TOKENLESS_PRODUCTION)) /
91              100;
92      }
93      return veBoostedLpBalance;
94  }
```



Per the README.md:

The Gauge/Market will receive rewardTokens from SCY as well as claiming the PENDLE token from gauge controller. All of the reward tokens (including PENDLE) will be distributed with boosting mechanism

While it's natural and reasonable for the PENDLE rewards to be distributed according to the `activeBalance`, we find it's quite unconventional and troublesome if this also applies to the distribution of SCY external rewards.

Firstly and philosophically, we would say that the external rewards is bonded to the SCY, therefore, they should not be redistributed according to any other metrics besides the `lpBalance`.

Furthermore and more practically, the redistribution mechanism will effectively make the calculation of the profit-and-loss much more complicated for the users.

That's because by the same time they get PENDLE rewards, they may gain more external rewards or lose part of their initial external rewards based on the amount of `veBalance`, in comparison to just hold their SCY tokens or the external rewards are distributed based on `lpBalance`.

Status

 Acknowledged



WP-G7: `vePENDLE.sol` Combine two external calls into one can save gas

Gas

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/PendleGauge.sol#L83-L84>

```
83     uint256 vePendleBalance = vePENDLE.balanceOf(user);
84     uint256 vePendleSupply = vePENDLE.totalSupplyCurrent();
```

Recommendation

Consider adding a new function called `totalSupplyAndBlanaceCurrent` in `vePENDLE` :

```
1  function totalSupplyAndBlanaceCurrent(address user) external view returns
   (uint128, uint128) {
2      (VeBalance memory supply, ) = _applySlopeChange();
3
4      uint128 userBalance = balanceOf(user);
5      return (userBalance, supply.getCurrentValue());
6  }
```

Status

✓ Fixed



WP-G8: Avoid unnecessary storage read can save gas

Gas

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/VotingEscrow/VotingEscrowPendleMainchain.sol#L191-L210>

```
191     function _applySlopeChange() internal returns (VeBalance memory, uint128) {
192         VeBalance memory supply = _totalSupply;
193         uint128 wTime = lastSlopeChangeAppliedAt;
194         uint128 currentWeekStart = WeekMath.getCurrentWeekStart();
195
196         if (wTime >= currentWeekStart) {
197             return (supply, wTime);
198         }
199
200         while (wTime < currentWeekStart) {
201             wTime += WEEK;
202             supply = supply.sub(slopeChanges[wTime], wTime);
203             totalSupplyAt[wTime] = supply.getValueAt(wTime);
204         }
205
206         _totalSupply = supply;
207         lastSlopeChangeAppliedAt = wTime;
208
209         return (supply, lastSlopeChangeAppliedAt);
210     }
```

At L209, use `wTime` instead of `lastSlopeChangeAppliedAt` can avoid unnecessary storage read and save some gas.

Status

✓ Fixed



WP-I9: End the **Vote** a few hours earlier before the next week starts can help avoid leftover

Informational

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/GaugeController/PendleGaugeControllerBaseUpg.sol#L117-L128>

```
117 function _addRewardsToMarket(address market, uint128 pendleAmount) internal {
118     MarketRewardData memory rwd = _getUpdatedMarketReward(market);
119     uint128 leftover = (rwd.incentiveEndsAt - rwd.lastUpdated) * rwd.pendlePerSec;
120     uint128 newSpeed = (leftover + pendleAmount) / WEEK;
121
122     rewardData[market] = MarketRewardData({
123         pendlePerSec: newSpeed,
124         accumulatedPendle: rwd.accumulatedPendle,
125         lastUpdated: uint128(block.timestamp),
126         incentiveEndsAt: uint128(block.timestamp) + WEEK
127     });
128 }
```

In the current implementation, the voting for allocation to the pools in the next week/epoch only ends by the start of the next week/epoch.

After the voting ends and the new epoch/week starts, the voting results can be synced to the side chains through a crosschain messaging protocol.

However, this will take some time, depends on when the sync can be triggered and how long it takes for the cross-chain message to be processed. This creates a gap between the end of the last reward period and the start of a new reward period, during that gap period, there will be no rewards.

An alternative solution would be to end the vote earlier, say a few hours, to allow the voting results to be synced sooner, and avoid the gap.



Recommendation

To apply this alternative solution, a decent amount of code changes would be required, and even with this early result synchronisation mechanism implemented, we still can't guarantee no gap.

Therefore, we believe it's not unnecessary to make any changes for this. And we leave this issue as it is for your reference only.

Status

ⓘ Acknowledged



WP-I10: `VotingEscrowPendleMainchain#_broadcast()` should refund the unspent crosschain message fee

Informational

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/VotingEscrow/VotingEscrowPendleMainchain.sol#L238-L245>

```
238     function _broadcast(  
239         uint256 chainId,  
240         uint256 wTime,  
241         VeBalance memory supply,  
242         bytes memory userData  
243     ) internal {  
244         _sendMessage(chainId, abi.encode(wTime, supply, userData));  
245     }
```

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/CelerAbstracts/CelerSenderUpd.sol#L28-L33>

```
28     function _sendMessage(uint256 chainId, bytes memory message) internal {  
29         assert(sidechainContracts.contains(chainId));  
30         address toAddr = sidechainContracts.get(chainId);  
31         uint256 fee = celerMessageBus.calcFee(message);  
32         celerMessageBus.sendMessage{ value: fee }(toAddr, chainId, message);  
33     }
```

When a crosschain message is needed, the caller is required to pay for the message with `msg.value` .

However, in the current implementation, the unspent part (`msg.value - fee`) will not be refunded to `msg.sender` .



This is not a problem with the current implementation of Celer's

`MessageBusSender.sol#calcFee()` :

<https://github.com/celer-network/sgn-v2-contracts/blob/d20dfaed94019c0404af0c86fce6ccb4c71b4b0d/contracts/message/messagebus/MessageBusSender.sol#L108-L127>

```
108  /**
109      * @notice Calculates the required fee for the message.
110      * @param _message Arbitrary message bytes to be decoded by the destination app
111      * contract.
112      * @return The required fee.
113      */
114  function calcFee(bytes calldata _message) public view returns (uint256) {
115      return feeBase + _message.length * feePerByte;
116  }
117
118
119  function setFeePerByte(uint256 _fee) external onlyOwner {
120      feePerByte = _fee;
121      emit FeePerByteUpdated(feePerByte);
122  }
123
124  function setFeeBase(uint256 _fee) external onlyOwner {
125      feeBase = _fee;
126      emit FeeBaseUpdated(feeBase);
127  }
```

Because the result of `calcFee()` will not change for the same message unless the owner changed the `feeBase` or `feePerByte` .

We observed the recent transactions on Celer's `MessageBusSender.sol` contract and it seems they rarely change them.

It may not continue to be so if Celer changes the way they update the `feeBase` or `feePerByte` .

Plus, we may choose to use another cross chain messaging provider later, which may have a more dynamic messaging fee.



Therefore, we still recommend you to add the logic to refund the unspend fee to the caller.

Status

✓ Fixed



WP-I11: Adding mainchain `gaugeController` into `sidechainContracts` is confusing and unnecessary

Informational

Issue Description

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/VotingController/PendleVotingControllerUpg.sol#L231-L263>

```
231  function _broadcastResults(  
232      uint64 chainId,  
233      uint128 wTime,  
234      uint128 totalPendlePerSec  
235  ) internal {  
236      uint256 totalVotes = weekData[wTime].totalVotes;  
237      if (totalVotes == 0) return;  
238  
239      uint256 length = chainPools[chainId].length();  
240      if (length == 0) return;  
241  
242      address[] memory pools = chainPools[chainId].values();  
243      uint256[] memory totalPendleAmounts = new uint256[](length);  
244  
245      for (uint256 i = 0; i < length; ++i) {  
246          uint256 poolVotes = weekData[wTime].poolVotes[pools[i]];  
247          uint256 pendlePerSec = (totalPendlePerSec * poolVotes) / totalVotes;  
248          totalPendleAmounts[i] = pendlePerSec * WEEK;  
249      }  
250  
251      if (chainId == block.chainid) {  
252          address gaugeController = sidechainContracts.get(chainId);  
253          IPGaugeControllerMainchain(gaugeController).updateVotingResults(  
254              wTime,  
255              pools,  
256              totalPendleAmounts  
257          );  
258      } else {  
259          _sendMessage(chainId, abi.encode(wTime, pools, totalPendleAmounts));  
260      }
```



```
261
262     emit BroadcastResults(chainId, wTime, totalPendlePerSec);
263 }
```

Per the README.md:

On Ethereum, there is a `VotingController` contract to control the voting on incentives for the different markets on the different chains.

When `chainId == block.chainid`, it means that we are `updateVotingResults()` for the current network, the "mainchain".

The current implementation requires the mainchain's `gaugeController` address to be added to `sidechainContracts`, this may cause some misunderstandings as it's not explicit how the name implies.

For example, `getAllSidechainContracts()` will also return mainchain's `gaugeController` address:

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/6b7ec5e22cc07617fd531dc71b1b2d2031aa68fe/contracts/core/LiquidityMining/CelerAbstracts/CelerSenderUpg.sol#L46-L58>

```
46  function getAllSidechainContracts()
47      public
48      view
49      returns (uint256[] memory chainIds, address[] memory addrs)
50  {
51      uint256 length = sidechainContracts.length();
52      chainIds = new uint256[](length);
53      addrs = new address[](length);
54
55      for (uint256 i = 0; i < length; ++i) {
56          (chainIds[i], addrs[i]) = sidechainContracts.at(i);
57      }
58  }
```



Recommendation

Consider adding a immutable variable `gaugeControllerMainchain` to store the mainchain's `gaugeController` address and only include sidechain contracts in `sidechainContracts` .

Status

✓ Fixed



WP-I12: Users will **not** naturally update others' *activeBalance*

Informational

Issue Description

vePENDLE Whitepaper assumes that **Users will naturally update others' activeBalance** :

6.3.1 Users will naturally update others' activeBalance

If there is no new locked position, the activeBalance of a user go down with every update. As such, a rational user u will go around calling the update function on everyone else except for u (assuming gas is not a concern), which will increase u 's boosting compared to others. As a result, everyone's activeBalance will be updated frequently, reflecting closely the real-time non- cached value.

We find that it's more complicated than that:

<https://github.com/pendle-finance/pendle-core-internal-v2/blob/ffcd33ca4103a22bfc4d1a1821a5d14e5dda4f59/contracts/core/Market/PendleMarket.sol#L233-L235>

```
233  function redeemRewards(address user) external nonReentrant returns (uint256[]  
    memory) {  
234      return _redeemRewards(user);  
235  }
```

While it's true that anyone can call **PendleMarket#redeemRewards()** to update other user's **activeBalance** .

The user's **activeBalance** won't necessarily go down with every update even if there is no new locked position.

The **activeBalance** of the user will then be updated based on this formula:

$$activeBalance_u = \min(lpBalance_u, boostedBalance_u)$$



$$boostedBalance_u = 0.4lpBalance_u + 0.6totalLP \frac{vePendleValue_u^Y}{veTotalSupply^Y}$$

In the chart below, we demonstrated how the $\frac{vePendleValue_u^Y}{veTotalSupply^Y}$ changes over time for the two users (to simplify, we assume there are only 2 users) with different unlock time and there is no new locked position:

- Orange dash line: User 1's `vePendleBalance`;
- Green dash line: User 2's `vePendleBalance`;
- Red dotted line: `veTotalSupply`
- Purpule solid line: User 1's $\frac{vePendleValue_u^Y}{veTotalSupply^Y}$
- Blue solid line: User 2's $\frac{vePendleValue_u^Y}{veTotalSupply^Y}$

<https://www.desmos.com/calculator/cfvs7ujr9f>

As you can see in the chart, user 2's $\frac{vePendleValue_u^Y}{veTotalSupply^Y}$ gets higher over time, and therefore their `activeBalance` will actually go up with every update.

User 1's `activeBalance`, in the other hand, will indeed go down with every update.

That's because the `activeBalance` of each user is not strictly related to the user's `vePendleBalance`, but rather related to the $\frac{vePendleValue_u^Y}{veTotalSupply^Y}$.


For a particular user, while even though their `vePendleBalance` is decreasing, if the `veTotalSupply` is decreasing even faster, then their `activeBalance` will actually go up.

As a very sophisticated user, they may go around and calculate the delta amount of `activeBalance` for the other users and only call the update function when their `activeBalance` is going down.

Before they can do that, the caller needs to gather a list of the users to be updated first. Seems like there is no easy way to get the list from on-chain data. So they might need to get the list from the events instead.

Besides, the gas costs to update others' `activeBalance` can often be higher than the potential gains. Therefore, it requires the caller to calculate and benefit and compare it with the real time gas cost to decide whether to update the `activeBalance` for a certain user or not.

To conclude, we believe a regular user or even a professional investor, may not have the



motivation (significant amount of gains) and skills/tools needed to actively update others'
`activeBalance` .

Resolution

The claim has been removed from the whitepaper.

Status

✓ Fixed



Appendix

Timeliness of content

The content contained in the report is current as of the date appearing on the report and is subject to change without notice, unless indicated otherwise by WatchPug; however, WatchPug does not guarantee or warrant the accuracy, timeliness, or completeness of any report you access using the internet or other means, and assumes no obligation to update any information following publication.



Disclaimer

This report is based on the scope of materials and documentation provided for a limited review at the time provided. Results may not be complete nor inclusive of all vulnerabilities. The review and this report are provided on an as-is, where-is, and as-available basis. You agree that your access and/or use, including but not limited to any associated services, products, protocols, platforms, content, and materials, will be at your sole risk. Smart Contract technology remains under development and is subject to unknown risks and flaws. The review does not extend to the compiler layer, or any other areas beyond the programming language, or other programming aspects that could present security risks. A report does not indicate the endorsement of any particular project or team, nor guarantee its security. No third party should rely on the reports in any way, including for the purpose of making any decisions to buy or sell a product, service or any other asset. To the fullest extent permitted by law, we disclaim all warranties, expressed or implied, in connection with this report, its content, and the related services and products and your use thereof, including, without limitation, the implied warranties of merchantability, fitness for a particular purpose, and non-infringement. We do not warrant, endorse, guarantee, or assume responsibility for any product or service advertised or offered by a third party through the product, any open source or third-party software, code, libraries, materials, or information linked to, called by, referenced by or accessible through the report, its content, and the related services and products, any hyperlinked websites, any websites or mobile applications appearing on any advertising, and we will not be a party to or in any way be responsible for monitoring any transaction between you and any third-party providers of products or services. As with the purchase or use of a product or service through any medium or in any environment, you should use your best judgment and exercise caution where appropriate. FOR AVOIDANCE OF DOUBT, THE REPORT, ITS CONTENT, ACCESS, AND/OR USAGE THEREOF, INCLUDING ANY ASSOCIATED SERVICES OR MATERIALS, SHALL NOT BE CONSIDERED OR RELIED UPON AS ANY FORM OF FINANCIAL, INVESTMENT, TAX, LEGAL, REGULATORY, OR OTHER ADVICE.